

## Overview

Introducing the first Potters and Makers collab project – **Make Your Own Pottery Tools**. We've designed this to be as simple as possible however that doesn't mean you won't get stuck or have questions. Since most of the steps here are relevant to the Makers group, check out the LMCS calendar for session times and join in the fun.

### Summary of Steps

1. Choose a pottery tool to make. At this point we can cater for either a chop or spinner.
2. Pick your poison! There are several different software packages that you can use for modelling. Have a chat to a Maker for guidance.
3. Model your Stamp or Spinner and generate your STL file.
4. Slice your model and generate the g-code. Cura is recommended.
5. Use the 3D printer to print!

## Modelling Software

Below is a list of the software packages that are in use at LMCS, each with its own pros and cons. Your choice will come down to a balance of:

- Ability – how quickly you can learn a new program.
- Budget – free or paid; and
- Time and frequency – how much time you have and how often you're likely to create models.

	Difficulty	Capability	Cost	Installed at the Shed	Account Required?
<a href="#">SketchUp</a>	**	***	Free (limited)	Yes (web-based)	Yes
<a href="#">OpenSCAD</a>	***	**	Free	Yes	No
<a href="#">Fusion 360</a>	*****	*****	Free (limited)	Yes (web-based)	Yes
<a href="#">Solidworks</a>	****	*****	~ \$50 / year	No	Yes



From the four software programs above, OpenSCAD is the quickest to get going. We have provided some sample scripts for you to use in the Appendix of this document which can be modified to suit your needs. The Shed computers have OpenSCAD already installed with the scripts available for use.

# OpenSCAD

## Steps

1. Download and install OpenSCAD from this link: <https://openscad.org/index.html>
2. Insert the relevant script from the appendix into the "editor" window once you've got OpenSCAD up and running. Render (F6) and then Export to STL (F7).
3. Save exported STL file to the LMCS Google Drive or onto a USB stick so that you can access it at the Shed if you need to.

Example screenshots of the scripts below.



## Slicing


Cura is recommended for slicing (preparing your STL file for 3D printing).

### Ultimaker Cura

#### Steps

1. Download and install Cura from this link: <https://ultimaker.com/software/ultimaker-cura/>
  - a. You can run Cura without setting up an account.
  - b. Cura is already installed on the LMCS computers.
2. Set up the printer as **Crealitty Ender-3 / Ender-3 v2** and the material (i.e. the filament) as **eSun PLA+ Black**.
3. Open your STL file.
4. Change the Print Settings according to how adventurous you are. More advanced settings give you more control of the outcome. **Slice** and **Preview** the results and tweak the settings until you're happy with the results.
5. If Cura can't produce a good enough result you may need to go back to your model (SCAD or otherwise) and select a different font (or change some other variables such as the dimensions).
6. Save the exported g-code file to the LMCS Google Drive or onto a USB stick so that you can access it at the Shed if you need to. From there you can put the code onto the SD card and insert into one of the 3D printers for printing!

Here is an example screenshot of what happens when you slice your STL stamp file in Cura. The slider on the very right of the screen allows you to see each layer of the print; a great way to verify your print beforehand.



## Assemble or Claim Your Makers Mark

If you let a Pottery Subcommittee member know what your stamp is, they can make a record and it helps ensure your creations can always be identified. For the spinner, there are some spare bearings in return for a donation to the Shed.

## Appendix

### OpenSCAD Script – Pottery Stamp (Text)

Using the script below make sure you review each of the variables to refine your design before printing it. You may wish to save on plastic (for example) and create a much shorter stamp so that you can attach your own (wooden) handle.

----- Copy text below this line -----

```
// This script can be used to create a pottery stamp
// Instructions:
// 1. Change the variables below to suit your needs
// 2. Click on the 'preview' button (F5)
// 3. Check the resulting model. Make changes if you need to.
// 4. Click on the 'render' button (F6)
// 5. Click on the 'export as STL' button (F7)

// Variables you can change
$fn = 100; // The higher this number is, the more smooth edges are.
stampHeight = 50; // Height of your stamp in mm
stampWidth = 10; // Width of your stamp in mm
stampLength = 10; // Length (or second width) of your stamp in mm
stampLetter = ":-)"; // The stamp letter(s)
stampFont = "Stencil"; // The font of your letter(s)
stampLetterScale = 0.8; // Scaling factor for the letter(s)
stampLetterHeight = 3; // How deep the letter stamps (height of the letter(s))

// You can also play/change with the code below if you're adventurous

// This line of code creates the base of the stamp
translate([0, 0, stampHeight/2])
  cube([stampWidth, stampLength, stampHeight], center = true);

// This line of code creates the stamp letter(s)
#mirror([0,1,0])
  translate([0,0,stampHeight])
    scale([stampLetterScale, stampLetterScale, 1])
      linear_extrude(stampLetterHeight)
        text(text = stampLetter, font = stampFont, halign = "center", valign =
"center");
```

----- Copy text above this line -----

## OpenSCAD Script – Pottery Stamp (Image)

With the script below you'll need to reference an "SVG" file. Did you know that Office 365 gives you access to 8000 stock images? Open PowerPoint -> Insert an Icon -> Right-click on the inserted picture and "Save as Picture".

We also have a script that will work with PNG files. This may give you some more flexibility however be aware the rendering is more resource intensive and therefore can be a bit more difficult to get right.

----- Copy text below this line -----

```
// This script can be used to create a pottery stamp
// Instructions:
// 1. Change the variables below to suit your needs
// 2. Click on the 'preview' button (F5)
// 3. Check the resulting model. Make changes if you need to.
// 4. Click on the 'render' button (F6)
// 5. Click on the 'export as STL' button (F7)

// Variables you can change
stampHeight = 50;           // Height of your stamp in mm
stampWidth = 10;           // Width of your stamp in mm
stampLength = 10;         // Length (or second width) of your stamp in mm
stampLetterScale = 0.1;    // Scaling factor for the image
stampLetterHeight = 2;     // Depth of the stamp
stampSVGFile = "image.svg"; // The stamp image name and location

// You can also play/change with the code below if you're adventurous

// This line of code creates the base of the stamp
translate([0, 0, stampHeight/2])
  cube([stampWidth, stampLength, stampHeight], center = true);

// This line of code creates the stamp image
#mirror([0,1,0])
  translate([0,0,stampHeight])
    scale([stampLetterScale, stampLetterScale, 1])
      linear_extrude(stampLetterHeight)
        import(file = stampSVGFile, center = true, dpi = 96);
```

----- Copy text above this line -----

## OpenSCAD Script – Trimming Spinner

Note that the script below will 3D print the spinner in a single print. This means you'll need to cut and separate the two pieces once the print has finished. Alternatively, you can separate the code below into two prints.

----- Copy text below this line -----

```
// This script can be used to create a pottery trimming spinner.
// Instructions:
// 1. Change the variables below to suit your needs
// 2. Click on the 'preview' button (F5)
// 3. Check the resulting model. Make changes if you need to.
// 4. Click on the 'render' button (F6)
// 5. Click on the 'export as STL' button (F7)

// Variables are based on a bearing of type 608RS (skateboard bearing)
bir = 3.95; //Bearing inner radius is 4mm
bor = 11.1; //Bearing outer radius is 11mm
bt = 7; //Bearing thickness (height)
wt = 2; //Trimmer wall thickness
tt = 4; //Trimmer top thickness
th = 15; //Trimmer top height
ch = 0.5; //Chamfer
rh = 1; //Recess height
rw = 1.5; //Recess width

//Trimmer Top
difference(){
color("steelblue")
    translate([0,0,wt+rh])
        rotate_extrude($fn = 200)
            polygon(
points=[[0,0],[bir,0],[bir,bt],[bir+rw,bt],[bir+rw,bt+rh],[bor+wt,bt+rh],[bor+wt,bt+rh+wt],[tt+wt,bt+rh+wt+th],[0,bt+rh+wt+th]] );

        //This bit of code puts the indents into the top of the spinner
        translate([0,0,bt+rh+wt+th-3])
            make_ring_of(bor+wt+1, 3){
                sphere(7, $fn = 100);
            }
}

//Trimmer Base
color("red")
    rotate_extrude($fn = 200)
        polygon( points=[[0,0],[bor+wt-
ch,0],[bor+wt,ch],[bor+wt,bt+wt+rh],[bor,bt+wt+rh],[bor,wt+rh],[bor-rw,wt+rh],[bor-
rw,wt],[0,wt]] );

module make_ring_of(radius, count){
    for (a = [0 : count - 1]) {
        angle = a * 360 / count;
        translate(radius * [cos(angle), -sin(angle), 0])
            children();
    }
}
```

----- Copy text above this line -----